Wolfram Cloud Einführung

Fahrzeugmodell

Wir schreiben die Fahrzeug Differentialgleichung (DGL)

In[29]:=
```
Proc = v'[t] == -b / m v[t] + Fu / m
```

Out[29]=
$$v'[t] == \frac{Fu}{m} - \frac{b\ v[t]}{m}$$

Werte der Standardparameter festlegen

In[30]:=
```
dfltParams = {b → 50, m → 1000, v0 → 5}
```

Out[30]=
```
{b → 50, m → 1000, v0 → 5}
```

DLG lösen

In[31]:=
```
ProcSol = DSolve[{Proc, v[0] == v0}, v[t], t]
```

Out[31]=
$$\left\{\left\{v[t] \to \frac{e^{-\frac{b\,t}{m}}\left(-Fu + e^{\frac{b\,t}{m}}\,Fu + b\,v0\right)}{b}\right\}\right\}$$

In[32]:=
```
ProcSolVal = ProcSol /. Append[dfltParams, Fu → 500]
```

Out[32]=
$$\left\{\left\{v[t] \to \frac{1}{50}\,e^{-t/20}\left(-250 + 500\,e^{t/20}\right)\right\}\right\}$$

In[33]:=
```
Plot[v[t] /. ProcSolVal, {t, 0, 100}, PlotRange → {0, 11}, AxesLabel → {t, v[t]}]
```

Out[33]=



P-Regler

Wechsel in 'regelungstechnische' Variablennamen:

In[34]:=
```
Procy = Proc /. {v → y, Fu → u[t]}
```

Out[34]=
$$y'[t] == \frac{u[t]}{m} - \frac{b\, y[t]}{m}$$

Definition des Stellgesetzes:

In[35]:=
```
Pctrl = u[t] → K * (w[t] - y[t])
```

Out[35]=
$$u[t] \to K\,(w[t] - y[t])$$

Definition der Führungsgröße

In[36]:=
```
RefVal = 5 + 3 * HeavisideTheta [t]
```

Out[36]=
$$5 + 3\,\text{HeavisideTheta}\,[t]$$

Einsetzen und lösen:

In[37]:=
```
CtrlLoop = Procy /. Pctrl /. w[t] → RefVal
CtrlLoopSol = DSolve[{CtrlLoop , y[0] == 5}, y[t], t]
```

Out[37]=
$$y'[t] == \frac{K\,(5 + 3\,\text{HeavisideTheta}\,[t] - y[t])}{m} - \frac{b\, y[t]}{m}$$

Out[38]=
$$\left\{\left\{y[t] \to \frac{5\, e^{-\frac{b\,t}{m} - \frac{K\,t}{m}} \left(b + e^{\frac{(b+K)\,t}{m}} K\right)}{b + K} + \left(-\frac{5\, e^{-\frac{b\,t}{m} - \frac{K\,t}{m}} \left(b + e^{\frac{(b+K)\,t}{m}} K\right)}{b + K} + \frac{e^{-\frac{b\,t}{m} - \frac{K\,t}{m}} \left(5\,b - 3\,K + 8\, e^{\frac{(b+K)\,t}{m}} K\right)}{b + K}\right) \text{HeavisideTheta}\,[t]\right\}\right\}$$

Standardparameter einsetzen:

In[39]:=
```
CtrlLoopSolVal = CtrlLoopSol /. dfltParams
FullSimplify [%]
```

Out[39]=
$$\left\{\left\{y[t] \to \frac{5\, e^{-\frac{t}{20} - \frac{K\,t}{1000}} \left(50 + e^{\frac{(50+K)\,t}{1000}} K\right)}{50 + K} + \left(-\frac{5\, e^{-\frac{t}{20} - \frac{K\,t}{1000}} \left(50 + e^{\frac{(50+K)\,t}{1000}} K\right)}{50 + K} + \frac{e^{-\frac{t}{20} - \frac{K\,t}{1000}} \left(250 - 3\,K + 8\, e^{\frac{(50+K)\,t}{1000}} K\right)}{50 + K}\right) \text{HeavisideTheta}\,[t]\right\}\right\}$$

Out[40]=
$$\left\{\left\{y[t] \to \frac{K\,(5 + 3\,\text{HeavisideTheta}\,[t]) + e^{-\frac{(50+K)\,t}{1000}} (250 - 3\,K\,\text{HeavisideTheta}\,[t])}{50 + K}\right\}\right\}$$

An dieser Stelle bitte die Verstärkung K variieren:

In[41]:=
```
Plot[{RefVal , y[t] /. CtrlLoopSolVal /. K → 500}, {t, 0, 10}, PlotRange → {0, 10}]
```

Out[41]=



Bitte berechnet die Stellgröße.

PI-Regler

Gleichungen des Reglers festlegen

In[50]:=
```
PIctrl = {x1 '[t] == w[t] - y[t],  u[t] == Kp * (w[t] - y[t]) + Ki * x1[t]}
```

Out[50]=
$$\{x1'[t] == w[t] - y[t],\ u[t] == Ki\ x1[t] + Kp\ (w[t] - y[t])\}$$

Prozess und Regler verknüpfen

In[51]:=
```
ODEs = Join[{Procy}, PIctrl]
```

Out[51]=
$$\left\{y'[t] == \frac{u[t]}{m} - \frac{b\ y[t]}{m},\ x1'[t] == w[t] - y[t],\ u[t] == Ki\ x1[t] + Kp\ (w[t] - y[t])\right\}$$

Anfangsbedingungen setzen

In[52]:=
```
ICs = {y[0] == 5,  x1[0] == 0}
```

Out[52]=
$$\{y[0] == 5,\ x1[0] == 0\}$$

Alles außer Regler Parameter ineinander einsetzen:

In[53]:=
```
ODEsMostVal = Join[ODEs , ICs] /. dfltParams /. w[t] → 8
```

Out[53]=
$$\left\{y'[t] == \frac{u[t]}{1000} - \frac{y[t]}{20},\ x1'[t] == 8 - y[t],\ u[t] == Ki\ x1[t] + Kp\ (8 - y[t]),\ y[0] == 5,\ x1[0] == 0\right\}$$

In[54]:=
```
Sol = DSolve[ODEsMostVal , {y[t], x1[t], u[t]}, t];
```

Regel Parameter setzen und plotten:

In[69]:=
```
SolVal = Sol /. {Ki → 700 , Kp → 600};
```

In[70]:=
```
U = FullSimplify [u[t] /. SolVal]
PU = Plot[ U, {t, 0, 20}, PlotRange → {0, 4000}];

Y = FullSimplify [y[t] /. SolVal]
PY = Plot[{Y, RefVal}, {t, 0, 20}, PlotRange → {0, 10}];

X1 = FullSimplify [x1[t] /. SolVal]
PX1 = Plot[{X1}, {t, 0, 20}, PlotRange → {0, 20}];

GraphicsRow [{PU, PY, PX1}]
```

Out[70]=
$$\left\{400 + \frac{200}{951} e^{-13\, t/40} \left(6657 \cos\left[\frac{\sqrt{951}\ t}{40}\right] + 325\ \sqrt{951}\ \sin\left[\frac{\sqrt{951}\ t}{40}\right]\right)\right\}$$

Out[72]=
$$\left\{8 + \frac{e^{-13\, t/40}\left(-3\ \sqrt{951}\ \cos\left[\frac{\sqrt{951}\ t}{40}\right] + 23 \sin\left[\frac{\sqrt{951}\ t}{40}\right]\right)}{\sqrt{951}}\right\}$$

Out[74]=
$$\left\{\frac{4}{7} + \frac{4\, e^{-13\, t/40}\left(-951 \cos\left[\frac{\sqrt{951}\ t}{40}\right] + 197\ \sqrt{951}\ \sin\left[\frac{\sqrt{951}\ t}{40}\right]\right)}{6657}\right\}$$

Out[76]=